Méthode d'Euler

II - Calcul formel et représentation graphique

```
> with(plots):
Warning, the name changecoords has been redefined
```

```
> Eq:=diff(y(x),x)-sin(x)*y(x)=x^2;
  sol:=rhs(dsolve({Eq,y(0)=a},y(x)));
```

$$Eq := \left(\frac{d}{dx}\,\mathrm{y}(x)\right) - \sin(x)\,\mathrm{y}(x) = x^2$$

$$sol := \mathbf{e}^{(-\cos(x))} \int_0^x \_z1^2\, \mathbf{e}^{\cos(\_z1)}\, d\_z1 + \frac{\mathbf{e}^{(-\cos(x))}\, a}{\mathbf{e}^{(-1)}}$$

On remarque la solution proposée n'est pas explicite car elle fait apparaître une intégrale. Cette solution est issue de la méthode de la variation de la constante.

```
> L:=[seq(sol,a=-5..5)]: # On génère la liste des expressions des solutions à tracer.
  plot(L,x=0..2,color=black);
  graphe1:=%:
```



III- Description de la méthode d'euler

```
> h=(b-a)/n;
```

$$h = \frac{b-a}{n}$$

```
> x[i]=a+i*h;
```

$$x_i = a + i\,h$$

Soit g la solution de (E) telle que g(x_i)=y_i
On a alors :

```
> D(g)(x)=f(x,g(x));
```

$$D(g)(x)=f(x, g(x))$$

Donc :

```
> D(g)(x[i])=f(x[i],g(x[i]));
  D(g)(x[i])=f(x[i],y[i]);
```

$$D(g)(x_i)=f(x_i, g(x_i))$$

$$D(g)(x_i)=f(x_i, y_i)$$

L'équation de la tangente de g en x_i est donc :

```
> y=f(x[i],y[i])*(x-x_i)+y[i];
```

$$y=f(x_i, y_i)\,(x-x\_i)+y_i$$

La solution F approchée proposée par la méthode d'Euler admet donc pour expression sur [x_i,x_(i+1)]

```
> F(x)=f(x[i],y[i])*(x-x[i])+y[i];
```

$$F(x)=f(x_i, y_i)\,(x-x_i)+y_i$$

Or :

```
> y[i]=F(x[i+1]);
```

$$y_i=F(x_{i+1})$$

D'où :

```
> y[i]=f(x[i],y[i])*(x[i+1]-x[i])+y[i];
```

$$y_i=f(x_i, y_i)\,(x_{i+1}-x_i)+y_i$$

Or :

```
> x[i+1]-x[i]=h;
```

$$x_{i+1}-x_i=h$$

D'où :

```
> y[i]=h*f(x[i],y[i])+y(i);
```

$$y_i=h\,f(x_i, y_i)+y(i)$$

IV- Implémentation de la méthode

```
> Euler:=proc(y0,n,a,b,f)
         local xi,yi,S,i,h;
  xi:=a;
  yi:=y0;
  S:=[a,y0];
  h:=evalf((b-a)/n);
  for i from 0 to n-1 do
     # xi et yi contiennent x_i et y_i
     yi:=evalf(h*f(xi,yi)+yi);
     xi:=evalf(xi+h);
     # xi et yi contiennent x_(i+1) et y_(i+1)
```

```
    S:=S,[xi,yi]; # On ajoute le point construit à la séquence
  od;
  [S];
  end;
```

$Euler := \mathbf{proc}(y0, n, a, b, f)$

$\mathbf{local}\ \xi, yi, S, i, h;$

$\xi := a;$

$yi := y0;$

$S := [a, y0];$

$h := \mathrm{evalf}((b-a)/n);$

$\mathbf{for}\ i\ \mathbf{from}\ 0\ \mathbf{to}\ n-1\ \mathbf{do}\ yi := \mathrm{evalf}(h * f(\xi, yi) + yi);\ \xi := \mathrm{evalf}(\xi + h);\ S := S, [\xi, yi]$

$\mathbf{end\ do};$

$[S]$

$\mathbf{end\ proc}$

L'équation (E') est équivalente à :

```
> diff(y(x),x)=sin(x)*y(x)+x^2;
```

$$\frac{d}{dx}\mathrm{y}(x) = \sin(x)\,\mathrm{y}(x) + x^2$$

On pose donc :

```
> f:=(x,y)-> sin(x)*y+x^2;
```

$$f := (x, y) \rightarrow \sin(x)\, y + x^2$$

```
> approx:=Euler(1,20,0,2,f):
  sol1:=subs(a=1,sol):
  plot([approx,sol1],x=0..2, color=[red,black]);
```



V - Etude comparative

```
> L:=NULL:
  for k from -5 to 5 do
      L:=L, Euler(k,20,0,2,f);
  od:
  plot([L],color=red);
  graphe2:=%:
```




```
> display({graphe1,graphe2});
```



Nous pouvons constater visuellement que l'approximation est bonne à proximité de 0, c'est-à-dire au niveau de la condition initiale et que la précision se détèriore au fur et à mesure que l'on parcours l'intervalle.

Nous répétons l'opération en augmentant n à 100.

```
> L:=NULL:
  for k from -5 to 5 do
      L:=L, Euler(k,100,0,2,f);
  od:
  graphe2:=plot([L],color=red):
  display({graphe1,graphe2});
```



Nous observons le même phénomène.
Cependant, nous observons que l'augmentationde n à augmenter la précision de l'approximation.